/*
 *  Copyright (c) 2010 Brian Stanek <bstanek@gmail.com>
 *
 *  Permission is hereby granted, free of charge, to any person
 *  obtaining a copy of this software and associated documentation
 *  files (the "Software"), to deal in the Software without
 *  restriction, including without limitation the rights to use,
 *  copy, modify, merge, publish, distribute, sublicense, and/or sell
 *  copies of the Software, and to permit persons to whom the
 *  Software is furnished to do so, subject to the following
 *  conditions:
 *
 *  The above copyright notice and this permission notice shall be
 *  included in all copies or substantial portions of the Software.
 *
 *  THE SOFTWARE IS PROVIDED "AS IS", WITHOUT WARRANTY OF ANY KIND,
 *  EXPRESS OR IMPLIED, INCLUDING BUT NOT LIMITED TO THE WARRANTIES
 *  OF MERCHANTABILITY, FITNESS FOR A PARTICULAR PURPOSE AND
 *  NONINFRINGEMENT. IN NO EVENT SHALL THE AUTHORS OR COPYRIGHT
 *  HOLDERS BE LIABLE FOR ANY CLAIM, DAMAGES OR OTHER LIABILITY,
 *  WHETHER IN AN ACTION OF CONTRACT, TORT OR OTHERWISE, ARISING
 *  FROM, OUT OF OR IN CONNECTION WITH THE SOFTWARE OR THE USE OR
 *  OTHER DEALINGS IN THE SOFTWARE.
 */

package guice.binding
{

import guice.Binder;
import guice.Binding;
import guice.Injector;
import guice.Key;
import guice.Provider;
import guice.scopes.Scoping;
import guice.spi.BindingScopingVisitor;
import guice.spi.BindingTargetVisitor;
import guice.spi.ElementVisitor;
import guice.spi.InstanceBinding;

public class AbstractBinding implements Binding
{

    //--------------------------------------------------------------------------
    //
    //  Constructor
    //
    //--------------------------------------------------------------------------

    public function AbstractBinding(source:Object, key:Key, scoping:Scoping,
            injector:Injector=null, factory:Factory=null)
    {
        super();
        this.source = source;
        this.key = key;
        this.scoping = scoping;
        this.injector = injector;
        this.factory = factory;
    }

    //--------------------------------------------------------------------------
    //
    //  Variables
    //
    //--------------------------------------------------------------------------

    private var factory:Factory;

    private var injector:Injector;

    private var key:Key;

    private var provider:Provider;

    private var scoping:Scoping;

    private var source:Object;

    //--------------------------------------------------------------------------
    //
    //  Abstract Methods: Binding
    //
    //--------------------------------------------------------------------------

    public function acceptTargetVisitor(visitor:BindingTargetVisitor):Boolean
    {
        return false;
    }

    public function applyTo(binder:Binder):void
    {
    }

    //--------------------------------------------------------------------------
    //
    //  Methods: Binding
    //
    //--------------------------------------------------------------------------

    public function acceptScopingVisitor(visitor:BindingScopingVisitor):Boolean
    {
        return scoping.acceptVisitor(visitor);
    }

    public function acceptVisitor(visitor:ElementVisitor):Boolean
    {
        return visitor.visitBinding(this);
    }

    public function getKey():Key
    {
        return key;
    }

    public function getProvider():Provider
    {
        if (provider == null)
        {
            if (injector == null)
            {
                throw new Error("getProvider() not supported for module bindings");
            }
            provider = injector.getProviderForKey(key);
        }
        return provider;
    }

    public function getSource():Object
    {
        return source;
    }

    //--------------------------------------------------------------------------
    //
    //  Methods
    //
    //--------------------------------------------------------------------------

    public function getFactory():Factory
    {
        return factory;
    }

    public function getInjector():Injector
    {
        return injector;
    }

    public function getScoping():Scoping
    {
        return scoping;
    }

    public function isConstant():Boolean
    {
        return this is InstanceBinding;
    }

    public function withScoping(scoping:Scoping):AbstractBinding
    {
        throw new Error();
    }

    public function withKey(key:Key):AbstractBinding
    {
        throw new Error();
    }

}

}
